

```
1| #! /usr/bin/env raku
2|
3| # Get the Pod vs. Code structure of a Raku/Pod6 file.
4| # © 2023 Shimon Bollinger. All rights reserved.
5| # Last modified: Tue 05 Sep 2023 09:39:29 PM EDT
6| # Version 0.0.1
7|
8| # no-weave
9| # always use the latest version of Raku
10| use v6.*;
11| use PrettyDump;
12| use Data::Dump::Tree;
13| # end-no-weave
```

**A grammar to parse a file into Pod and Code sections.**

# INTRODUCTION

I want to create a semi-literate Raku source file with the extension `.s1`. Then, I will *weave* it to generate a readable file in formats like Markdown, PDF, HTML, and more. Additionally, I will *tangle* it to create source code without any Pod6.

To do this, I need to divide the file into Pod and Code sections by parsing it. For this purpose, I will create a dedicated Grammar.

## Convenient tokens

Let's create three tokens for convenience.

```
14| #      We need to declare them with C<my> because we
15| #      need to use them in a subroutine later. #TODO explain why.
16|
17|      my token rest-of-line {      \N* [\n | $]  }
18|      my token ws-till-EOL  {      \h* [\n | $]  }
19|      my token blank-line   {      ^^ <ws-till-EOL> }
```

# The Grammar

Our file will exclusively consist of Pod or Code sections, and nothing else. The Code sections are of two types, a) code that is woven into the documentation, and b) code that is not woven into the documentation. The TOP token clearly indicates this.

```
20| #use Grammar::Tracer;
21| grammar Semi::Literate is export {
22|     token TOP { [ <pod> | <non-woven-code> | <woven-code>]* }
```

## The Pod6 delimiters

According to the [documentation](#),

**Every Pod6 document has to begin with =begin pod 1 and end with =end pod.**

So let's define those tokens.

### The begin token

```
23|     my token begin {
24|         ^^ \h* \= begin <.ws> pod
```

Most programming applications do not focus on the structure of the executable file, which is not meant to be easily read by humans. Our tangle would replace all the Pod6 blocks with a single `\n`. That can clump code together that is easier read if there were one or more blank lines.

However, we can provide the option for users to specify the number of empty lines that should replace a pod block. To do this, simply add a number at the end of the `=begin` directive. For example, `=begin pod. [1]`

```
25|         [ \h* $<num-blank-lines>=(\d+) ]? # an optional number to specify the
26|                                             # number of blank lines to replace the
27|                                             # C<Pod> blocks when tangling.
```

The remainder of the `begin` directive can only be whitespace.

```
28|         <ws-till-EOL>
29|     } # end of my token begin
```

### The end token

The end token is much simpler.

```
30|     my token end { ^^ \h* \= end <.ws> pod <ws-till-EOL> }
```

## The Pod token

Within the delimiters, all lines are considered documentation. We will refer to these lines as `plain-lines`. Additionally, it is possible to have nested Pod sections. This allows for a hierarchical organization of documentation, allowing for more structured and detailed explanations.

It is also permissible for the block to be empty. Therefore, we will use the 'zero-or-more' quantifier on the lines of documentation, allowing for the possibility of having no lines in the block.

```
31|     token pod {
32|         <begin>
```

---

[1] This is non-standard Pod6 and will not compile until woven!

```

33|          [<pod> | <plain-line>]*
34|          <end>
35|      } # end of token pod

```

## The Code tokens

The Code sections are similarly easily defined. There are two types of Code sections, depending on whether they will appear in the woven code. See [below](#) for why some code would not be included in the woven code.

### Woven sections

These sections are trivially defined. They are just one or more `plain-lines`.

```

36|      token woven-code { <plain-line>+ }

```

### Non-woven sections

Sometimes there will be code you do not want woven into the document, such as boilerplate code like `use v6.d;`. You have two options to mark such code. By individual lines or by delimited blocks of code.

```

37|      token non-woven-code {
38|          | <one-line-no-weave>+
39|          | <delimited-no-weave>+
40|      } # end of token non-woven

```

### One line of code

Simply append `#no-weave` at the end of the line!

```

41|      token one-line-no-weave {
42|          ^^ \N*
43|          '#' <.ws> 'no-weave'
44|          <.ws> <rest-of-line>
45|      } # end of token one-line-no-weave

```

### Delimited blocks of code

Simply add comments `#no-weave` and `#end-no-weave` before and after the code you want ignored in the formatted document.

```

46|      token delimited-no-weave {
47|          <begin-no-weave>
48|          <plain-line>*?
49|          <end-no-weave>
50|      } # end of token delimited-no-weave
51|
52|      token begin-no-weave {
53|          ^^ \h*                                # optional leading whitespace
54|          '#' <.ws> 'no-weave'                    # the delimiter itself (#no-weave)
55|          <.ws> <rest-of-line>                    # optional trailing whitespace or comment
56|      } # end of token <begin-no-weave>
57|
58|      token end-no-weave {
59|          ^^ \h*                                # optional leading whitespace
60|          '#' <.ws> 'end-no-weave'                # the delimiter itself (#end-no-weave)
61|          <.ws> <rest-of-line>                    # optional trailing whitespace or comment

```

```
62|     } # end of token <end--no-weave>
```

## The plain-line token

The plain-line token is, really, any line at all...

```
63|     token plain-line {  
64|         $<plain-line> = [^^ <rest-of-line>]
```

## Disallowing the delimiters in a plain-line.

... except for one subtlety. They it can't be one of the begin/end delimiters. We can specify that with a Regex Boolean Condition Check.

```
65|         <?{ &not-a-delimiter($<plain-line>.Str) }>  
66|     } # end of token plain-line
```

This function simply checks whether the plain-line match object matches either the begin or end token.

Incidentally, this function is why we had to declare those tokens with the my keyword. This function wouldn't work otherwise.

```
67|     sub not-a-delimiter (Str $line --> Bool) {  
68|         return not $line =~ /<begin> | <end>/;  
69|     } # end of sub not-a-delimiter (Match $line --> Bool)
```

And that concludes the grammar for separating Pod from Code!

```
70| } # end of grammar Semi::Literate
```

# The Tangle subroutine

This subroutine will remove all the Pod6 code from a semi-literate file (.sl) and keep only the Raku code.

```
71| #TODO multi sub to accept Str & IO::PatGh
72| sub tangle (
```

The subroutine has a single parameter, which is the input filename. The filename is required. Typically, this parameter is obtained from the command line or passed from the subroutine MAIN.

```
73|     Str $input-file!,
```

The subroutine will return a Str, which will be a working Raku program.

```
74|     --> Str ) is export {
```

First we will get the entire Semi-Literate .sl file...

```
75|     my Str $source = $input-file.IO.slurp;
```

## Clean the source

### Remove unnecessary blank lines

Very often the code section of the Semi-Literate file will have blank lines that you don't want to see in the tangled working code. For example:

```
sub foo () {
    { ... }
} # end of sub foo ()

# <== unwanted blank lines
# <== unwanted blank lines

# <== unwanted blank lines
# <== unwanted blank lines
```

So we'll remove the blank lines immediately outside the beginning and end of the Pod6 sections.

```
76|     $source ~~ s:g/\=end (\N*)\n+/\=end$0\n/;
77|     $source ~~ s:g/\n+\=begin /\n\=begin/;
```

## The interesting stuff

We parse it using the Semi::Literate grammar and obtain a list of submatches (that's what the caps method does) ...

```
78|     my Pair @submatches = Semi::Literate.parse($source).caps;
```

...and iterate through the submatches and keep only the code sections...

```
79|     my Str $raku-code = @submatches.map( {
80|         when .key eq 'woven-code' | 'non-woven-code' {
81|             .value;
82|         }
```

## Replace Pod6 sections with blank lines

```
83|         when .key eq 'pod' {
84|             my $num-blank-lines = .value.hash<begin><num-blank-lines>;
85|             "\n" x $num-blank-lines with $num-blank-lines;
86|         }
87|
88|         #no-weave
89|         default { die "Tangle: should never get here. .key == {.key}" }
90|         #end-no-weave
```

... and we will join all the code sections together...

```
91|     } # end of my Str $raku-code = @submatches.map(
92|     ).join;
```

## Remove the *no-weave* delimiters

```
93|     $source ~~ s:g{ ^^ \h* '#' <.ws> 'no-weave' <rest-of-line> } = '';
94|     $source ~~ s:g{ ^^ (.*) '#' <.ws> 'no-weave' <rest-of-line> } = "$0\n";
95|     $source ~~ s:g{ ^^ \h* '#' <.ws> 'end-no-weave' <rest-of-line> } = '';
```

## remove blank lines at the end

```
96|     $raku-code ~~ s{\n <blank-line>* $ } = '';
```

And that's the end of the tangle subroutine!

```
97|     return $raku-code;
98| } # end of sub tangle (
```



# The Weave subroutine

The Weave subroutine will *weave* the `.sl` file into a readable Markdown, HTML, or other format. It is a little more complicated than `sub tangle` because it has to include the code sections.

```
99| sub weave (
```

## The parameters of Weave

`sub weave` will have several parameters.

### `$input-file`

The input filename is required. Typically, this parameter is obtained from the command line through a wrapper subroutine `MAIN`.

```
100|     Str $input-file!;
```

### `$format`

The output of the weave can (currently) be Markdown, Text, or HTML. It defaults to Markdown. The variable is case-insensitive, so 'markdown' also works.

```
101|     Str :f(:$format) is copy = 'markdown';
102|     #= The output format for the woven file.
```

### `$line-numbers`

It can be useful to print line numbers in the code listing. It currently defaults to True.

```
103|     Bool :l(:$line-numbers) = True;
104|     #= Should line numbers be added to the embeded code?
```

`sub weave` returns a `Str`.

```
105|     --> Str ) is export {
```

`#TODO`

```
106|     my UInt $line-number = 1;
```

First we will get the entire `.sl` file...

```
107|     my Str $source = $input-file.IO.slurp;
108|
109|     my Str $cleaned-source;
110|
111|     $cleaned-source = $source;
112|     #=begin pod 1
113|     #
114|     #=head3 Remove full comment lines followed by blank lines
115|     #
116|     #=end pod
117|     #
118|     #   # delete full comment lines
119|     #   $source ~~ s:g{ ^^ \h* '#' \N* \n+} = '';
```

```

120| #
121| #   # remove Raku comments, unless the '#' is escaped with
122| #   # a backslash or is in a quote. (It doesn't catch all quote
123| #   # constructs...(that's a TODO))
124| #   # And leave the newline.
125| #
126| #=begin pod 1
127| #
128| #=head3 Remove EOL comments
129| #
130| #=end pod
131| #
132| #   for $source.split("\n") -> $line {
133| #       my $m = $line ~~ m{
134| #           ^^
135| #           $<stuff-before-the-comment> = ( \N*? )
136| #
137| #           #TODO make this more robust - allow other delimiters, take into
138| #           #account the Q language, heredocs, nested strings...
139| #           <!after          # make sure the '#' isn't in a string
140| #               ( [
141| #                   | \\\
142| #                   | \" <- [\" ]>*
143| #                   | \' <- [\' ]>*
144| #                   | \[] <- [\[]>*
145| #               ] )
146| #           >
147| #           "#"
148| #
149| #
150| #           # We need to keep these delimiters.
151| #           # See the section above "Remove code marked as 'no-weave'".
152| #           <!before
153| #               [
154| #                   | 'no-weave'
155| #                   | 'end-no-weave'
156| #               ]
157| #           >
158| #           \N*
159| #           $$ };
160| #
161| #       $cleaned-source ~= $m ?? $<stuff-before-the-comment> !! $line;
162| #       $cleaned-source ~= "\n";
163| #   } # end of for $source.split("\n") -> $line
164| #

```

## Remove blank lines at the begining and end of the code

### EXPLAIN THIS!

```

165| $cleaned-source ~~ s:g{\=end (\N*)\n+} = "\=end$0\n";
166| $cleaned-source ~~ s:g{\n+\=begin (<.ws> pod) [<.ws> \d]?} = "\n\=begin$0";

```

## Interesting stuff ...Next, we parse it using the `Semi::Literate` grammar and obtain a list of submatches (that's what the `caps` method does) ...

```
167| my Pair @submatches = Semi::Literate.parse($cleaned-source).caps;
```

...And now begins the interesting part. We iterate through the submatches and insert the code sections into the Pod6...

```
168| my Str $weave = @submatches.map( {  
169|     when .key eq 'pod' {  
170|         .value  
171|     } # end of when .key
```

#TODO

```
172|         when .key eq 'woven-code' { qq:to/EOCB/; }  
173|             \=begin pod  
174|             \=begin code :lang<raku>  
175|                 { my $fmt = ($line-numbers ?? "%3s| " !! ' ') ~ "%s\n";  
176|                     .value  
177|                     .lines  
178|                     .map($line-numbers  
179|                         ?? {"%4s| %s\n".sprintf($line-number++, $_) }  
180|                         !! { "%s\n".sprintf(          $_) }  
181|                     )  
182|                     .chomp;  
183|                 }  
184|             \=end code  
185|             \=end pod  
186|             EOCB  
187|  
188|         when .key eq 'non-woven-code' {  
189|             ; # do nothing  
190|         } # end of when .key eq 'non-woven'  
191|  
192|         # no-weave  
193|         default { die "Weave: should never get here. .key == {.key}" }  
194|         # end-no-weave  
195|     } # end of my $weave = Semi::Literate.parse($source).caps.map  
196|     ).join;
```

remove useless Pod directives

```
197| $weave ~~ s:g{ \h* \=end <.ws> pod <rest-of-line>  
198|             \h* \=begin <.ws> pod <rest-of-line> } = '';
```

## remove blank lines at the end

```
199| $weave ~~ s{\n <blank-line>* $ } = '';
```

And that's the end of the `tangle` subroutine!

```
200| return $weave  
201| } # end of sub weave (
```

# NAME

Semi::Literate - A semi-literate way to weave and tangle Raku/Pod6 source code.

# **VERSION**

This documentation refers to Semi-Literate version 0.0.1

# **SYNOPSIS**

```
use Semi::Literate;
```

```
# Brief but working code example(s) here showing the most common usage(s)
```

```
# This section will be as far as many users bother reading
```

```
# so make it as educational and exemplary as possible.
```

# DESCRIPTION

`Semi::Literate` is based on Daniel Sockwell's `Pod::Literate` module

A full description of the module and its features. May include numerous subsections (i.e. `=head2`, `=head2`, etc.)

# **BUGS AND LIMITATIONS**

There are no known bugs in this module. Patches are welcome.



# AUTHOR

Shimon Bollinger (deoac.bollinger@gmail.com)

# LICENSE AND COPYRIGHT

© 2023 Shimon Bollinger. All rights reserved.

This module is free software; you can redistribute it and/or modify it under the same terms as Raku itself. See [The Artistic License 2.0](#).

This program is distributed in the hope that it will be useful, but WITHOUT ANY WARRANTY; without even the implied warranty of MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.

```
202| # no-weave
203| my %*SUB-MAIN-OPTS =
204|     :named-anywhere,          # allow named variables at any location
205|     :bundling,                # allow bundling of named arguments
206|     # :coerce-allomorphs-to(Str), # coerce allomorphic arguments to given type
207|     :allow-no,                # allow --no-foo as alternative to --/foo
208|     :numeric-suffix-as-value, # allow -j2 as alternative to --j=2
209| ;
210|
211| #| Run with option '--pod' to see all of the Pod6 objects
212| multi MAIN(Bool :$pod!) is hidden-from-USAGE {
213|     for $=pod -> $pod-item {
214|         for $pod-item.contents -> $pod-block {
215|             $pod-block.raku.say;
216|         }
217|     }
218| } # end of multi MAIN (:$pod)
219|
220| #| Run with option '--doc' to generate a document from the Pod6
221| #| It will be rendered in Text format
222| #| unless specified with the --format option. e.g.
223| #|     --doc --format=HTML
224| multi MAIN(Bool :$doc!, Str :$format = 'Text') is hidden-from-USAGE {
225|     run $*EXECUTABLE, "--doc=$format", $*PROGRAM;
226| } # end of multi MAIN(Bool :$man!)
227|
228| my $semi-literate-file =
'/Users/jimbollinger/Documents/Development/raku/Projects/Semi-Literate/source/Literate.sl';
229| multi MAIN(Bool :$testt!) {
230|     say tangle($semi-literate-file);
231| } # end of multi MAIN(Bool :$testt!)
232|
233| multi MAIN(Bool :$testw!) {
234|     say weave($semi-literate-file);
235| } # end of multi MAIN(Bool :$testw!)
236|
237| #end-no-weave
```