

```
1| #! /usr/bin/env raku
2|
3| # Get the Pod vs. Code structure of a Raku/Pod6 file.
4| # © 2023 Shimon Bollinger. All rights reserved.
5| # Last modified: Sat 09 Sep 2023 04:32:24 PM EDT
6| # Version 0.0.1
7|
```

A grammar to parse a file into Pod and Code sections.

INTRODUCTION

I want to create a semi-literate Raku source file with the extension `.sl`. Then, I will *weave* it to generate a readable file in formats like Markdown, PDF, HTML, and more. Additionally, I will *tangle* it to create source code without any Pod6.

Convenient tokens

Let's create four tokens for convenience.

8	my token hws	{	<!ww>\h*	}	# Horizontal White Space
9	my token rest-of-line	{	\N* [\n \$]	}	
10	my token blank-line	{	^^ <ws-till-EOL>	}	

To do this, I need to divide the file into Pod and Code sections by parsing it. For this purpose, I will create a dedicated Grammar.

The Grammar

```
11| #use Grammar::Tracer;
12| grammar Semi::Literate is export {
```

Our file will exclusively consist of Pod or Code sections, and nothing else. The Code sections are of two types, a) code that is woven into the documentation, and b) code that is not woven into the documentation. The TOP token clearly indicates this.

```
13|     token TOP {
14|         [
15|             | <non-woven-code>
16|             | <pod>
17|             | <woven-code>
18|         ]*
19|     } # end of token TOP
```

The Pod6 delimiters

According to the [documentation](#),

Every Pod6 document has to begin with =begin pod and end with =end pod.

So let's define those tokens.

The begin-pod token

```
20|     token begin-pod {
21|         ^^ <hws> '=' begin <hws> pod <ws-till-EOL>
22|     } # end of token begin-pod
```

The end-pod token

The end-pod token is much simpler.

```
23|     token end-pod { ^^ <hws> '=' end <hws> pod <ws-till-EOL> }
```

Replacing Pod6 sections with blank lines Most programming applications do not focus on the structure of the executable file, which is not meant to be easily read by humans. Our tangle would replace all the Pod6 blocks with a single `\n`. That can clump code together that is easier read if there were one or more blank lines.

However, we can provide the option for users to specify the number of empty lines that should replace a pod block. To do this, simply add a Pod6 comment immediately after the `=begin pod` statement. The comment can say anything you like, but must end with a digit specifying the number of blank lines with which to replace the Pod6 section.

```
24|     token blank-line-comment {
25|         ^^ <hws>
26|         '=' comment
27|         \N*?
28|         $<num-blank-lines> = (\d+)?
29|         <ws-till-EOL>
30|     } # end of token comment
```

The Pod token

Within the delimiters, all lines are considered documentation. We will refer to these lines as `plain-lines`. Additionally, it is possible to have nested `Pod` sections. This allows for a hierarchical organization of documentation, allowing for more structured and detailed explanations.

It is also permissible for the block to be empty. Therefore, we will use the 'zero-or-more' quantifier on the lines of documentation, allowing for the possibility of having no lines in the block.

```

31|     token pod {
32|         <begin-pod>
33|         <blank-line-comment>?
34|         [<pod> | <plain-line>]*
35|         <end-pod>
36|     } # end of token pod

```

The Code tokens

The Code sections are similarly easily defined. There are two types of Code sections, depending on whether they will appear in the woven code. See [below](#) for why some code would not be included in the woven code.

Woven sections

These sections are trivially defined. They are just one or more plain-lines.

```
37| token woven-code { <plain-line>+ }
```

Non-woven sections

Sometimes there will be code you do not want woven into the document, such as boilerplate code like `use v6.0;`. You have two options to mark such code. By individual lines or by delimited blocks of code.

```

38|         token non-woven-code {
39|             [
40|                 | <one-line-no-weave>
41|                 | <delimited-no-weave>
42|             ]+
43|         } # end of token non-woven

```

One line of code

Simply append `# begin-no-weave` at the end of the line!

```
44|         token one-line-no-weave {
45|             ^^ \N*?
46|             '#' <hws> 'no-weave-this-line'
47|             <ws-till-EOL>
48|         } # end of token one-line-no-weave
```

Delimited blocks of code

Simply add comments `# begin-no-weave` and `#end-no-weave` before and after the code you want ignored in the formatted document.

```
49| token begin-no-weave {
50|     ^^ <hws>                # optional leading whitespace
```

```

51|         '#' <hws> 'begin-no-weave' # the delimiter itself (# begin-no-weave)
52|         <ws-till-EOL>              # optional trailing whitespace or comment
53|     } # end of token <begin-no-weave>
54|
55|     token end-no-weave {
56|         ^^ <hws>                    # optional leading whitespace
57|         '#' <hws> 'end-no-weave'    # the delimiter itself (#end-no-weave)
58|         <ws-till-EOL>              # optional trailing whitespace or comment
59|     } # end of token <end--no-weave>
60|
61|     token delimited-no-weave {
62|         <begin-no-weave>
63|         <plain-line>*
64|         <end-no-weave>
65|     } # end of token delimited-no-weave

```

The plain-line token

The plain-line token is, really, any line at all... .. except for one subtlety. They it can't be one of the begin/end delimiters. We can specify that with a Regex Boolean Condition Check.

```

66|     token plain-line {
67|         :my $*EXCEPTION = False;
68|         [
69|             || <begin-pod>          { $*EXCEPTION = True }
70|             || <end-pod>           { $*EXCEPTION = True }
71|             || <begin-no-weave>     { $*EXCEPTION = True }
72|             || <end-no-weave>       { $*EXCEPTION = True }
73|             || <one-line-no-weave> { $*EXCEPTION = True }
74|             || $<plain-line> = [^^ <rest-of-line>]
75|         ]
76|         <?{ !$*EXCEPTION }>
77|     } # end of token plain-line

```

And that concludes the grammar for separating Pod from Code!

```

78| } # end of grammar Semi::Literate

```

The Tangle subroutine

This subroutine will remove all the Pod6 code from a semi-literate file (.sl) and keep only the Raku code.

```
79| #TODO multi sub to accept Str & IO::PatGh
80| sub tangle (
```

```
81|     Str $input-file!,
```

The subroutine will return a Str, which will be a working Raku program.

```
82|         --> Str ) is export {
```

First we will get the entire Semi-Literate .sl file...

```
83|     my Str $source = $input-file.IO.slurp;
```

Clean the source

Remove unnecessary blank lines

Very often the code section of the Semi-Literate file will have blank lines that you don't want to see in the tangled working code. For example:

```
sub foo () {
    { ... }
} # end of sub foo ()

# <== unwanted blank lines
# <== unwanted blank lines

# <== unwanted blank lines
# <== unwanted blank lines
```

So we'll remove the blank lines immediately outside the beginning and end of the Pod6 sections.

```
84|     my Str $cleaned-source = $source;
85|     $cleaned-source ~~ s:g{\=end (\N*)\n+} = "\=end$0\n";
86|     $cleaned-source ~~ s:g{\n+\=begin (<hws> pod) [<hws> \d]?} = "\n\=begin$0";
```

The interesting stuff

We parse it using the Semi::Literate grammar and obtain a list of submatches (that's what the caps method does) ...

```
87|     my Pair @submatches = Semi::Literate.parse($cleaned-source).caps;
```

...and iterate through the submatches and keep only the code sections...

```
88| #     note "submatches.elems: {@submatches.elems}";
89|     my Str $raku-code = @submatches.map( {
90|         note .key;
91|         when .key eq 'woven-code' | 'non-woven-code' {
92|             .value;
93|         }
```

Replace Pod6 sections with blank lines

```
94|         when .key eq 'pod' {
95|             my $num-blank-lines =
96|                 .value.hash<blank-line-comment><num-blank-lines>;
97|             "\n" x $num-blank-lines with $num-blank-lines;
98|         }
99|
```

... and we will join all the code sections together...

```
100|     } # end of my Str $raku-code = @submatches.map(
101|     ).join;
```

Remove the *no-weave* delimiters

```
102|     $raku-code ~~ s:g{ ^^ <hws> '#' <hws> 'begin-no-weave'      <rest-of-line> }
103|         = '';
104|     $raku-code ~~ s:g{ ^^ <hws> '#' <hws> 'no-weave-this-line' <rest-of-line> }
105|         = "$0\n";
106|     $raku-code ~~ s:g{ ^^ <hws> '#' <hws> 'end-no-weave'          <rest-of-line> }
107|         = '';
```

remove blank lines at the end

```
108|     $raku-code ~~ s{\n <blank-line>* $ } = '';
```

And that's the end of the tangle subroutine!

```
109|     return $raku-code;
110| } # end of sub tangle (
```


The Weave subroutine

The Weave subroutine will *weave* the `.sl` file into a readable Markdown, HTML, or other format. It is a little more complicated than `sub tangle` because it has to include the code sections.

```
111| sub weave (
```

The parameters of Weave

`sub weave` will have several parameters.

`$input-file`

The input filename is required. Typically, this parameter is obtained from the command line through a wrapper subroutine `MAIN`.

```
112|     Str $input-file!;
```

`$format`

The output of the weave can (currently) be Markdown, Text, or HTML. It defaults to Markdown. The variable is case-insensitive, so 'markdown' also works.

```
113|     Str :f(:$format) is copy = 'markdown';
114|     #= The output format for the woven file.
```

`$line-numbers`

It can be useful to print line numbers in the code listing. It currently defaults to `True`.

```
115|     Bool :l(:$line-numbers) = True;
116|     #= Should line numbers be added to the embedded code?
```

```
117|     --> Str ) is export {
```

```
118|     my UInt $line-number = 1;
```

```
119|     my Str $source = $input-file.IO.slurp;
```

Remove blank lines at the begining and end of the code

EXPLAIN THIS!

```
120|     my Str $cleaned-source = $source;
121|     $cleaned-source ~~ s:g{\=end (\N*)\n+} = "\=end$0\n";
122|     $cleaned-source ~~ s:g{\n+\=begin (<hws> pod) [<hws> \d]?} = "\n\=begin$0";
```

Interesting stuff ...Next, we parse it using the `Semi::Literate` grammar and obtain a list of submatches (that's what the `caps` method does) ...

```
123|     my Pair @submatches = Semi::Literate.parse($cleaned-source).caps;
```

...And now begins the interesting part. We iterate through the submatches and insert the code sections into the Pod6...

```
124| #    note "weave submatches.elems: {@submatches.elems}";
125| #    note "submatches keys: {@submatches».keys}";
126|    my Str $weave = @submatches.map( {
127|        when .key eq 'pod' {
128|            .value
129|        } # end of when .key

130|            when .key eq 'woven-code' { qq:to/EOCB/; }
131|                \=begin pod
132|                \=begin code :lang<raku>
133|                    { my $fmt = ($line-numbers ?? "%3s| " !! ' ') ~ "%s\n";
134|                        .value
135|                        .lines
136|                        .map($line-numbers
137|                            ?? {"%4s| %s\n".sprintf($line-number++, $_) }
138|                            !! { "%s\n".sprintf(                $_) }
139|                        )
140|                        .chomp;
141|                    }
142|                \=end code
143|                \=end pod
144|                EOCB
145|
146|            when .key eq 'non-woven-code' {
147|                ''; # do nothing
148|            } # end of when .key eq 'non-woven'
149|

150|    } # end of my $weave = Semi::Literate.parse($source).caps.map
151|    ).join;
```

remove blank lines at the end

```
152|    $weave ~~ s{\n <blank-line>* $ } = '';
```

And that's the end of the tangle subroutine!

```
153|    return $weave
154| } # end of sub weave (
```

NAME

Semi::Literate - A semi-literate way to weave and tangle Raku/Pod6 source code.

VERSION

This documentation refers to Semi-Literate version 0.0.1

SYNOPSIS

```
use Semi::Literate;
```

```
# Brief but working code example(s) here showing the most common usage(s)
```

```
# This section will be as far as many users bother reading
```

```
# so make it as educational and exemplary as possible.
```

DESCRIPTION

`Semi::Literate` is based on Daniel Sockwell's `Pod::Literate` module

A full description of the module and its features. May include numerous subsections (i.e. `=head2`, `=head2`, etc.)

BUGS AND LIMITATIONS

There are no known bugs in this module. Patches are welcome.

AUTHOR

Shimon Bollinger (deoac.bollinger@gmail.com)

LICENSE AND COPYRIGHT

© 2023 Shimon Bollinger. All rights reserved.

This module is free software; you can redistribute it and/or modify it under the same terms as Raku itself. See [The Artistic License 2.0](#).

This program is distributed in the hope that it will be useful, but WITHOUT ANY WARRANTY; without even the implied warranty of MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.